A BASIC PROGRAM FOR ANALYSIS OF DROPLET SIZE DISTRIBUTION IN INSECTICIDE SPRAYS

L. R. BOOBAR, J. H. NELSON,
L. M. ANDERSON AND M. R. SARDELIS[1]

US Army Medical Bioengineering Research and Development Laboratory, Fort Detrick, Frederick, MD 21710–5010

Personal computers, by rapidly performing the repetitive mathematical operations involved in ultra low volume (ULV) aerosol droplet sample size distribution analysis, afford an opportunity to increase knowledge of insecticide sprays. Programs have been written for hand-held calculators (West and Cashman 1980) and personal computers (Sofield and Kent 1984) to calculate volume median diameter (VMD) for data from slide wave samples (Mount and Pierce 1972). The development of the US Army hot-wire droplet measuring device, model DC-2A, manufactured and marketed by KLD Associates (Mahler and Magnus 1984), produced a requirement for a program to calculate droplet size distribution and VMD from the data displayed by this device. A demonstration program (Fig. 1) is presented in IBM BASIC, and with minor syntax changes can be converted to other BASIC dialects.

The "VARIABLE SECTION" starting with line 125 (Fig. 1) is the first of three distinct sections in the program. This section allows the user to vary the descriptive characteristics of the sampling device. Many of these characteristics (i.e., sensing wire dimensions, number of bins into which the aerosol spectrum is divided, etc.) are used in the calculation of droplet size distribution and VMD. The flexibility permits this program to be used with other models of hot-wire droplet analyzers.

```
100  WIDTH "LPT1:",108:CLS
105  PRINT "This Program Calculates VmD and Size"
110  PRINT "Distribution for Aerosol Droplets"
115  PRINT "using Hot-Wire Instrumentation."
120 REM
125 REM                 "**** VARIABLE SECTION *****"
130 REM
135  DIM CD(25): DIM AP(25): DIM LL(25)
140  DIM MD(25): DIM MF(25): DIM ND(25)
145  DIM PC(25): DIM UL(25): DIM VD(25)
150  PRINT: PRINT
155  PRINT "Enter the following information."
160  PRINT: PRINT
165      INPUT "Diameter of sensing wire (microns)";D
170      INPUT "Length of sensing wire (centimeters)";L
175      INPUT "Interrupt time/droplet (seconds)";IT
180      INPUT "Number of bins";B
185  FOR I=1 TO B
190    PRINT "Lower limit of droplet diameter for bin",I
195    INPUT LL(I)
200    PRINT "Upper limit of droplet diameter for bin",I
205    INPUT UL(I)
210    PRINT "Multiplication factor for bin",I: INPUT MF(I)
215 REM "CD(I) = Change in diameter per bin"
220 REM "MD(I) = Median diameter in microns"
225    CD(I)= UL(I)-LL(I)
230    MD(I)= CD(I)/2 + LL(I)
235  NEXT I: CLS
240 REM "Initialization"
245  SN=0: SV=0: SP=0: W=0
250      INPUT "Collector's Name"; COLL$
255      INPUT "Temperature during sampling"; TEMP$
260      INPUT "Site"; SITE$
265      INPUT "Date"; DAT$
270      INPUT "Insecticide"; INS$
275      INPUT "Sprayer Type"; SPR$
280      INPUT "Sample Time(seconds)"; ST
285      INPUT "Flow velocity(cm/sec)";V
290  FOR I= 1 TO B
295    PRINT "Number of droplets in bin"I: INPUT ND(I)
300 REM "Sum of ND(I)"
305    SN= SN + ND(I)
310  NEXT I:CLS
315 REM "T=time,ST=sample time,SN*IT=total IT for sample"
320  T= ST - SN*IT
325 REM
330 REM               "**** MAIN PROGRAM SECTION *****"
335 REM
340  FOR I= 1 TO B
345 REM "VD is Volumetric Distribution in cubic microns"
350  VD(I)= .5236 *MD(I)^3*((MF(I)*ND(I))/(V*T*L*(MD(I)+D)))
355 REM "Sum of VD(I)"
360    SV= SV + VD(I)
365  NEXT I
370  FOR I= 1 TO B
375    PC(I)= VD(I)/SV * 100
380 REM "ACCUMULATIVE PC(I)"
385    SP = SP + PC(I)
390    AP(I)= SP
395  NEXT I
400  FOR I= 1 TO B
405    IF AP(I)=50 THEN VMD=MD(I):GOTO 430
410    IF AP(I)<50 THEN GOTO 425
415    VMD=MD(I-1) + CD(I)*(50- AP(I-1))/(AP(I)-AP(I-1))
420    GOTO 430
425  NEXT I
430  PRINT "                    U.L.V. DROPLET SIZE DISTRIBUTION ANALYSIS"
435  PRINT: PRINT "Collector:" COLL$ TAB(40)        "Date:" DAT$
440  PRINT "Site:" SITE$  TAB(40)                   "Temperature:" TEMP$
445  PRINT "Sprayer:" SPR$  TAB(40)                 "Sample Time:" ST
450  PRINT "Insecticide:" INS$: PRINT
455  PRINT "      Med. Droplet   No. of     Mult.     Vol.        % of
 Accum %"
460  PRINT "Bins    Diameter    Droplets    Factor    Dist.     Tot. Vol.
of Tot. Vol."
465  PRINT "(B)       (MD)        (ND)       (MF)      (VD)        (PC)
   (AP)"
470  FOR I = 1 TO B
475    PRINT USING "##        ###         #####      ##.#     ####.##       ##.##
        ###.##"; I, MD(I), ND(I), MF(I), VD(I), PC(I), AP(I)
480  NEXT I
485  PRINT "                          -----             ----------"
490  PRINT USING "                        ######            ######.##"; SN, SV
495  PRINT:PRINT USING "Volume Median Diameter = ##.# microns";VMD
500 REM
505 REM                 "**** REPORT SECTION *****"
510 REM
515  INPUT "Would you like a printout (Y/N)"; Y$
520    IF Y$ = "N" THEN GOTO 600:STOP
525    IF Y$ <> "Y" THEN GOTO 515
530  LPRINT TAB(22)        "U.L.V. DROPLET SIZE DISTRIBUTION ANALYSIS"
535  LPRINT:LPRINT TAB(5) "Collector:  " COLL$  TAB(60) "Date:  " DAT$
540  LPRINT TAB(5) "Site:  " SITE$  TAB(60)             "Temperature:  " TEMP$
545  LPRINT TAB(5) "Sprayer:  " SPR$  TAB(60)            "Sample Time: " ST
550  LPRINT TAB(5) "Insecticide:  " INS$:LPRINT:LPRINT
555  LPRINT TAB(12)      "Med. Droplet    No. of     Mult.      Vol.      % of
    Accum. % "
560  LPRINT TAB(5)      "Bins     Diameter     Droplets    Factor     Dist.   Tot
. Vol.   of Tot. Vol."
565  LPRINT TAB(5)          "(B)        (MD)          (ND)        (MF)       (VD)
   (PC)          (AP)"
570  FOR I =1 TO B
575    LPRINT USING            "    ##          ###          #####       ##.#
####.##      ##.##        ###.##";I, MD(I), ND(I), MF(I), VD(I), PC(I), AP(I)
580  NEXT I
585  LPRINT "                             -----              -------"
590  LPRINT USING "                           ######               ######.##";  SN,
 SV
595  LPRINT:LPRINT USING "     Volume Median Diameter = ##.# microns";VMD
600  CLS:INPUT "Do you have another sample to analyze (Y/N)";Y$
605    IF Y$ = "N" THEN GOTO 620:STOP
610    IF Y$ <> "Y" THEN GOTO 600
615    GOTO 245
620  END
```

Fig. 1. BASIC program for analysis of droplet size distribution.

Most variables are defined by the use of INPUT statements ending with the variable name. Variables not requiring input from the

[1] The opinions or assertions contained herein are the private views of the author(s) and are not to be construed as official or as reflecting the views of the Department of the Army or the Department of Defense.

keyboard are explained in remark (REM) statements. The only variables not located in this section are those used in the process of summing; however, these variables are also preceded by a REM statement. The following variables require additional explanation beyond that provided in the program.

Velocity (V), in line 285, represents air velocity at the tip of the probe during sampling. If the only information desired is VMD, which is based on an accumulative percentage (AP(I)) of 50, then V can be set at 1 cm/sec. Time (T) in line 315 is the difference between sample time (ST) and the total interruption time (IT). The DC-2A device, for example, has an IT equal to 2 msec; that is, at the moment a droplet impinges on the sensing wire (0.06 cm × 5.0 μ), a 2 msec inhibit interval is initiated during which time no droplet counts are accepted. Total IT varies with the number of droplets per sample.

The "MAIN PROGRAM SECTION" begins at line 330. Line 350 states the equation for calculating Volumetric Distribution, VD(I), in the ith size interval. Size intervals vary among bins but are always expressed in microns. The portion of the equation in parentheses calculates droplet concentration (drops/cm^3) from the number of droplets per bin, flow velocity, and collection time. All of these may be provided by a hot-wire droplet measuring device.

Printed records are controlled in the "REPORT SECTION," line 505. The report (Fig. 2) contains all necessary record-keeping information plus a detailed breakdown by bin of all important variables. A facsimile of the report appears on the screen to enable the user to decide if a printed report is desired. Line 600 gives the option of entering more samples. Since it takes time to enter and reenter individual hot-wire device characteristics, an IF-THEN instruction, also called a conditional transfer statement (Coan 1978), was placed into the program (Line 605) to bypass all nonvarying information during the processing of subsequent samples.

Hot-wire technology represents a significant advance in the methods by which aerosol clouds are sampled. Each device has individual design and operational characteristics. This program allows input of individual device characteristics with the intent of expanding use of the program beyond DC-2A data.

Manufacturers of hot-wire droplet measur-

U.L.V. DROPLET SIZE DISTRIBUTION ANALYSIS

Collector: SGT. Scott
Site: Ft. Detrick
Sprayer: G-88
Insecticide: Malathion

Date: 5/21/85
Temperature: 72
Sample Time: 100

Bins (B)	Med. Droplet Diameter (MD)	No. of Droplets (ND)	Aero. Factor (AF)	Vol. Dist. (VD)	% of Tot. Vol. (PC)	Accum. % of Tot. Vol. (AP)
1	1.0	6630	6.0	713.47	7.75	7.75
2	1.5	1882	4.4	462.69	5.03	12.77
3	2.5	378	2.5	211.86	2.30	15.08
4	6.5	274	1.3	915.37	9.94	25.02
5	12.5	186	1.0	2233.90	24.26	49.28
6	22.0	97	1.0	4116.57	44.71	93.99
7	31.5	6	1.0	552.90	6.01	100.00
8	40.0	0	1.0	0.00	0.00	100.00
9	90.0	0	1.0	0.00	0.00	100.00
10	170.0	0	1.0	0.00	0.00	100.00
11	200.0	0	1.0	0.00	0.00	100.00
12	200.0	0	1.0	0.00	0.00	100.00
13	200.0	0	1.0	0.00	0.00	100.00
14	200.0	0	1.0	0.00	0.00	100.00
		-----		-------		
		9453		9206.77		

Volume Median Diameter = 12.7 microns

Fig. 2. Sample program output.

ing devices plan to offer built-in printers for their machines, but the information provided will be limited to raw sample data and VMD. Therefore, programs for personal computers are necessary for more detailed analyses of droplet size distribution data.

The authors greatly appreciate the assistance given by Roy L. Scott in the preparation of this program.

References Cited

Coan, J. S. 1978. Basic basic. Hayden Book Co., Inc., Rochelle Park, NJ.

Mahler, D. S. and D. E. Magnus. 1984. Hot-wire technique for droplet measurements, p. 153–168 *In:* J. M. Tishkoff, R. D. Ingebo and J. B. Kennedy, (eds.) Liquid particle size measurement techniques. American Society for Testing and Materials Special Technical Publication 848, Philadelphia, PA.

Mount, G.A. and N. W. Pierce. 1972. Droplet size of ultra-low volume ground aerosol as determined by three collection methods. Mosq. News 32:586–589.

Sofield, R. K. and R. Kent. 1984. A basic program for the analysis of ULV insecticide droplets. Mosq. News 44:73–75.

West, H. W. and D. L. Cashman. 1980. The use of programmable calculators for the calculation of mass median diameter. Mosq. News 40:631–632.

PRESUMED *DIROFILARIA IMMITIS* INFECTIONS FROM FIELD-COLLECTED MOSQUITOES IN NORTH CAROLINA[1]

BEULAH M. PARKER

Department of Entomology, North Carolina State University, Raleigh, NC 27695–7613

Canine *Dirofilaria immitis* (Leidy) infection is prevalent in North Carolina (Rowley 1977[2], Butts 1979, Falls and Platt 1982), especially in coastal areas. A. R. Johnson (D.V.M., Countryview Animal Clinic, Bayboro, Pamlico County, NC, pers. commun.) has found the filarial worm in 80% of the 3-yr-old dogs examined from Hobucken, Pamlico County (35° 15′N latitude). Data are lacking on specific local vectors, but several species of mosquitoes are incriminated since they are among those (summarized by Buxton and Mullen 1980) found naturally infected with *D. immitis* in other regions of the U.S. However, the vector potential of a mosquito species may differ geographically (Christensen and Andrews 1976, Magnarelli 1978, Buxton and Mullen 1980). This investigation was conducted to determine potential mosquito vectors of *D. immitis* and extent of infections in a coastal area of North Carolina.

The study area was the community of Hobucken, located on South Goose Creek Island in northeast Pamlico County. Mosquitoes were collected in the yard of a residence near the edge of the salt marsh in southeast Hobucken and near a hunting dog kennel 3.2 km (2 mi.) inland from the edge of the salt marsh. The salt marsh was irregularly flooded, primarily by wind tides.

Adult female mosquitoes were collected on August 3 and 14, September 9 and 30, October 7, 15 and 28, and November 26 with CO_2 baited CDC light traps and by the human-bait method. Ninety percent of *Aedes sollicitans* (Walker), the most numerous species collected, were captured by the human-bait method and near the edge of the salt marsh. *Anopheles bradleyi* King was most abundant in light trap collections near the edge of the salt marsh. Collected mosquitoes were taken to the laboratory and stored at $-15°C$ until identified to species and examined for filarial worms.

After legs and wings were removed, the head, thorax and abdomen of each adult female mosquito were separated on a glass slide with the aid of a dissecting microscope, teased apart with insect pins in a drop of *Aedes* saline (Hayes 1953), then examined with 100 or 150X magnification of a compound microscope. The number of parasites in infected individuals was counted, recorded and measured with an ocular micrometer. Filarial worms were presumed to be *D. immitis* if their size range, structure, and developmental site in the mosquito were similar to those reported by Iyenger (1957) and Taylor (1960).

A total of 2,885 mosquitoes, comprising 10 species in 4 genera, were examined for filarial worms. Presumed *D. immitis* filarial worms were found in 19 (0.7%) of the mosquitoes examined. Four of 10 species of mosquitoes were parasitized (Table 1). The infective stage of the parasitic worm (L_3 larva) was found in 3 species, *Ae. sollicitans, Aedes taeniorhynchus* (Wiedemann), and *Culex salinarius* (Coquillett) (Table 2). Only first-stage filarial worms (within the Malpighian tubules) were found in parasitized *An. bradleyi* (Table 2). Infective stage larvae of *D. immitis* apparently have not been reported from field-collected *An. bradleyi*.

Natural infections of presumed *D. immitis*

---

[1] Paper No. 9912 of the Journal Series of the North Carolina Agricultural Research Service, Raleigh, NC 27695. This research was supported in part by NIH-BRSG No. PR07071.

[2] Rowley, B. J. 1977. The prevalence of heatworm, *Dirofilaria immitis* (Leidy, 1856), infection in privately owned and free-ranging dogs in Wake, Durham, and Orange counties, North Carolina. M. S. thesis, North Carolina State University, Raleigh. 22 pp.